

9/6.1

Analyse syntaxique d'une phrase

Nous avons vu comment créer les écrans d'un jeu d'aventures. De nombreuses étapes restent encore à franchir avant de pouvoir exécuter notre premier jeu.

Une des actions les plus importantes effectuée par un jeu d'aventures consiste à analyser les ordres entrés par le ou les joueurs.

Tous les jeux d'aventures capables d'interpréter un ordre ou une suite d'ordres entrés en toutes lettres au clavier mettent en œuvre un analyseur syntaxique. Comme son nom l'indique, un analyseur syntaxique est un programme qui analyse la syntaxe d'une phrase. Dans les cas conventionnels, le programme extrait des mots clés de la phrase à analyser, et vérifie si ces derniers peuvent être associés dans la salle actuelle.

Exemple :

La liste des mots clés est :

(PRENDRE, CASSER, AVANCER, NORD, SUD, EST, OUEST, BATON, CHANDELLE)

Le joueur entre la phrase :

« PRENDRE LE BATON QUI SE TROUVE SUR LA PIERRE »

Les mots clés identifiés sont **PRENDRE** et **BATON**.

Si, dans la salle où se trouve le joueur, le cas a été prévu, le programme agira en conséquence. Par exemple, il pourra répondre **« TU L'AS MAINTENANT DANS LES MAINS »**, et mémoriser que le joueur a un bâton en main pour la suite du jeu.

L'analyse syntaxique étudiée dans ce paragraphe portera sur les phrases ayant la structure suivante :

VERBE + SUJET

Il y aura, dans un premier temps, extraction des mots clés **VERBE** et **SUJET** ; dans un deuxième temps, analyse de la corrélation entre **VERBE** et **SUJET** en tenant compte de la salle où se trouve le joueur ; et enfin, dans un troisième temps, l'action découlant de la reconnaissance nulle, partielle ou totale de l'ordre entré par le joueur.

Phase 1 : Extraction de mots clés

Problème à résoudre

Soit une phrase de longueur n , et soit un mot clé de longueur p . Comment déterminer si le mot clé fait partie de la phrase ?


Résolution du problème

Parcourir la phrase de la première lettre à la $(n-p)$ ème lettre. Extraire à chaque fois un bloc de p lettres, et le comparer au mot clé. S'il y a similitude, c'est que le mot clé fait partie de la phrase.

Corrolaire

Si k mots clés sont définis, il faudra faire l'opération qui vient d'être décrite jusqu'à ce qu'un mot clé soit rencontré ou jusqu'à ce que les k mots clés aient été passés en revue.

Ce corrolaire peut être représenté comme suit :



Le problème que nous venons d'exposer peut être résolu par un programme Basic si le nombre de mots clés est faible, mais nous avons préféré présenter une version Assembleur de ce programme pour vous permettre de créer des jeux d'aventures complexes dans lesquels le temps d'analyse des phrases entrées par le joueur est très court.

Le listing du programme est le suivant :

```

1          ;
2          ;Analyse d'une phrase
3          ;Recherche de mots-cle
4          ;Entree: SCE=Pointeur Source
5          ;      TXT=Point. phrase entree
6          ;Sortie: NOCLE=No de mot-cle ou FF
7          ;
8          ;-----
9          ORG 9000H
10         LOAD 9000H
11         ;
12         NOCLE: DS 1          ;No Mot-cle
13         MOTCLE: DS 1        ;Nbre de Mots-cle
14         SCE: DS 2           ;Pointeur Source
15         TXT: DS 2           ;Pointeur phrase
16         STXT: DS 2          ;Sauvegarde TXT
17         ;
18 900B 2A0490 LD HL,(TXT)
19 900B 220690 LD (STXT),HL    ;Sauvegarde
20 900E 3E01 LD A,1
21 9010 320090 LD (NOCLE),A
22 9013 2A0290 LD HL,(SCE)
23 9016 7E LD A,(HL)          ;Nb de mots-cle
24 9017 320190 LD (MOTCLE),A    ;Sauvegarde
25 901A 23 INC HL
26         BX: EQU $
27 901B 220290 LD (SCE),HL    ;Mot cle No n

```

```

28 901E CD6590          CALL RECLen          ;Rech LEN(Mot-cle)
29 9021 0E00           LD  C,0              ;Rech en debut de phrase
30                    PO: EQU  $
31 9023 C5             PUSH BC              ;Sauvegarde
32 9024 2A0290         LD  HL,(SCE)
33 9027 ED5B0490       LD  DE,(TXT)
34 902B CD7090         CALL CMPARE          ;Comparaison
35 902E B7             OR  A
36 902F C1             POP  BC
37 9030 C8             RET  Z               ;Mot-cle identifie
38 9031 0C             INC  C
39 9032 79             LD  A,C
40 9033 FE28           CP  40
41 9035 2809           JR  Z,P2             ;Fin de comparaison
42 9037 2A0490         LD  HL,(TXT)
43 903A 23             INC  HL
44 903B 220490         LD  (TXT),HL        ;Prochain test
45 903E 18E3           JR  PO
46                    P2: EQU  $
47 9040 3A0190         LD  A,(MOTCLE)
48 9043 3D             DEC  A
49 9044 320190         LD  (MOTCLE),A
50 9047 2816           JR  Z,P3
51 9049 2A0690         LD  HL,(STXT)
52 904C 220490         LD  (TXT),HL        ;Restitution pointeur
53 904F 2A0290         LD  HL,(SCE)
54 9052 CD6590         CALL RECLen
55 9055 23             INC  HL
56 9056 3A0090         LD  A,(NOCLE)
57 9059 3C             INC  A

```

```

58 905A 320090          LD   (NOCLE),A
59 905D 18BC           JR   BX
60                   P3:   EQU  #
61 905F 3EFF           LD   A,OFFH           ;Aucun mot-cle identifie
62 9061 320090          LD   (NOCLE),A
63 9064 C9            RET
64                   ;-----
65                   ;Recherche de longueur de mot-cle
66                   ;-----
67                   ;Entree: HL=@ Mot-Cle
68                   ;Sortie: B =Longueur du mot-cle
69                   ;-----
70                   RECLN: EQU  #
71 9065 0600           LD   B,0
72 9067 05            DEC  B
73                   R1:   EQU  #
74 9068 04            INC  B
75 9069 7E            LD   A,(HL)
76 906A FEFF           CP   OFFH
77 906C C8            RET  Z           ;B=LEN(Mot-cle)
78 906D 23            INC  HL
79 906E 18FB          JR   R1
80                   ;-----
81                   ;Comparaison de chaines ASCII
82                   ;-----
83                   ;Entree: HL=@ Source
84                   ;      DE=@ Texte a comparer
85                   ;      B =Longueur comparaison
86                   ;Sortie: A =0 Si comparaison OK

```

```

87          ;          A =1 Sinon
88          ;-----
89          CMPARE:    EQU  $
90          CO:        EQU  $
91 9070 1A          LD   A,(DE)
92 9071 BE          CP   (HL)
93 9072 2006        JR   NZ,C1          ;Difference
94 9074 23          INC  HL
95 9075 13          INC  DE
96 9076 10F8        DJNZ CO          ;Boucle de comparaison
97 9078 1803        JR   C2          ;Comparaison OK
98          C1:      EQU  $          ;Comparaison ratee
99 907A 3E01        LD   A,1
100 907C C9         RET
101          C2:      EQU  $          ;Comparaison reussie
102 907D AF        XOR  A
103 907E C9         RET
104          END

```

Décomposition du programme

Initialisation Lignes 18 à 25
 Recherche de la longueur du mot clé n Ligne 28
 Test d'identification du mot clé n Lignes 29 à 45
 Passage au prochain mot clé Lignes 47 à 63
 Recherche de la longueur d'un mot clé Lignes 71 à 79
 Comparaison mot clé/portion de phrase Lignes 89 à 103

Supposons que les conditions suivantes aient été définies :

Salle	Verbe	Sujet	Changement de pièce	Réponse	Points	Fin
1	1	1	0	1	138	0
2	1	3	3	2	128	1
2	1	2	0	3	158	0

Soit l'ensemble des réponses suivant :

(TU L'AS DANS LES MAINS, UN SCORPION ETAIT CACHE DESSOUS, CA FAIT DU BIEN)

Analysons la signification du tableau ci-dessus :

- *Ligne 1* : Si le joueur se trouve en salle 1 et qu'il donne l'ordre (VERBE = 1, SUJET = 1), « **PRENDRE LAMPE** », aucun changement de salle n'est effectué.
L'ordinateur affiche (réponse 1) « **TU L'AS DANS LES MAINS** », le nombre de points de vie augmente de 10 (128 + 10), et la partie n'est pas finie (FIN = 0).
- *Ligne 2* : Si le joueur se trouve en salle 2 et qu'il donne l'ordre (VERBE = 1, SUJET = 3) « **PRENDRE BATON** », il y a passage en salle 3, l'ordinateur affiche « **UN SCORPION ETAIT CACHE DESSOUS** », le nombre de points de vie est inchangé (128), et la partie est terminée (FIN = 1).
- *Ligne 3* : Si le joueur se trouve en salle 2 et qu'il donne l'ordre (VERBE = 1, SUJET = 2) « **PRENDRE PO** », aucun changement de salle n'est effectué (Changt. = 0).
L'ordinateur affiche « **CA FAIT DU BIEN** », le nombre de points de vie augmente de 30 (128 + 30) et la partie peut se poursuivre (FIN = 0).

Remarque :

L'information « Points » (de vie) est un nombre signé sur 8 bits : par convention, 128 correspond à un accroissement nul du nombre de points de vie, $128 + n$ correspond à un accroissement de n du nombre de points de vie, et $128 - n$ à une diminution de n du nombre de points de vie.

Reprenons l'exemple précédent et voyons quel sera son codage en mémoire.

Rappel : Scènes (E1, E2, E3)
 Verbes (PRENDRE, VERSER)
 Sujets (LAMPE, PO, BATON)
 Réponses (TU L'AS DANS LES MAINS, UN SCORPION
 ETAIT CACHE DESSOUS, CA FAIT DU BIEN)

Codage mémoire

03/65/31/FF/65/32/FF/65/33/FF
 E 1 E 2 E 3

02/50/52/45/4E/44/52/45/FF/56/45/52/53/45/52/FF
 P R E N D R E V E R S E R

03/40/41/4D/50/45/FF/50/4F/FF/42/41/54/4F/4E/FF
 L A M P E P O B A T O N

03/54/55/20/40/27/41/53/20/44/41/4E/53/20/40/45/53/20/4D/41/49/4E/53/FF
 T U L ' A S D A N S L E S M A I N S

55/4E/20/53/43/4F/52/40/49/4F/4E/20/45/54/41/49/54/20/43/41/43/48/15
 U N S C O R P I O N E T A I T C A C H E

20/44/45/53/53/4F/55/53/FF/43/41/20/46/41/49/54/20/44/55/20
 D E S S O U S C A F A I T D U

42/49/45/4E/FF
 B I E N

03/01/01/01/00/01/8A/00/02/01/03/03/02/80/01/02/01/02/00/03/9E/00/FF
 C S V S C R P F S V S C R P F S V S C R P F
 o a e u h e t i a e u h e t i a e u h e t i
 n l r j g p s n l r j g p s n l r j g p s n
 d l b e t l b e t l b e t
 e e t e e t e e t


Maintenant qu'une convention est adoptée pour le codage en mémoire des divers éléments, voyons comment va s'organiser le programme d'analyse de la phrase.

1^{er} temps : Recherche de verbe ;

2^e temps : Recherche de sujet ;

3^e temps : Recherche de corrélation (Salle, Verbe, Sujet).

Cette décomposition du programme est clairement représentée par l'organigramme suivant :



La recherche de corrélation se fait comme suit :

- 1) Parcours de la table des conditions jusqu'à tomber sur une condition opérant sur la salle courante ;
- 2) Même démarche qu'au 1), mais concernant le verbe ;
- 3) Même démarche qu'au 1) mais concernant le sujet ;
- 4) Extraction des actions à effectuer (Changement de salle, réponse, modification du nombre de points de vie, fin du jeu).

Le listing du programme est le suivant :

```

1      ;Recherche de Verbe, Sujet et
2      ;d'une relation entre eux en
3      ;fonction de la salle courante
4      ;Entree: SALLE=No salle courante
5      ;      COND =@ Debut conditions
6      ;      SUJET=@ Debut sujets
7      ;      VERBE=@ Debut verbes
8      ;Sortie: CH=No de salle ou FF
9      ;      RE=No de reponse ou FF
10     ;      PT=Nbre de points ou FF
  
```

```

11          ;          FI=1 si fin de partie
12          ;-----
13          ORG  907FH
14          LOAD 907FH
15          SALLE: DS  1          ;No de salle
16          NCOND: DS  1          ;Nombre de conditions
17          COND:  DS  2          ;@ Debut conditions
18          VERBE: DS  2          ;@ Debut verbes
19          SUJET: DS  2          ;@ Debut sujets
20          SU:    DS  1          ;Sauv. Sujet
21          VE:    DS  1          ;Sauv. Verbe
22          CH:    DS  1          ;No de salle
23          RE:    DS  1          ;No de reponse
24          PT:    DS  1          ;Nombre de points
25          FI:    DS  1          ;Indic. fin de partie
26          NOCLE: EQU 9000H      ;No mot-cle
27          SCE:   EQU 9002H      ;Pointeur Source
28          TXT:   EQU 9004H      ;Pointeur Phrase
29          TEXTE: EQU 4300H      ;Debut phrase
30          RECHER: EQU 9008H     ;Pt entree Rech.
31          ;-----
32          ;RAZ Zone de Sortie
33 908D 218990          LD  HL,CH
34 9090 AF              XOR  A
35 9091 0604           LD  B,4
36          RAZ:      EQU  $
37 9093 77            LD  (HL),A
38 9094 23            INC  HL
39 9095 10FC          DJNZ RAZ

```

```

40      ; -----
41 9097 2A8390      LD  HL, (VERBE)
42 909A 220290      LD  (SCE),HL      ;Debut des verbes
43 909D 210043      LD  HL,TEXTE
44 90A0 220490      LD  (TXT),HL      ;Phrase utilisateur
45 90A3 CD0890      CALL RECHER      ;Recherche Verbe
46 90A6 3A0090      LD  A, (NOCLE)
47 90A9 328890      LD  (VE),A      ;Sauvegarde du verbe
48 90AC 2A8590      LD  HL, (SUJET)
49 90AF 220290      LD  (SCE),HL      ;Debut des Sujets
50 90B2 210043      LD  HL,TEXTE
51 90B5 220490      LD  (TXT),HL      ;Phrase Utilisateur
52 90B8 CD0890      CALL RECHER      ;Recherche Sujet
53 90BB 3A0090      LD  A, (NOCLE)
54 90BE 328790      LD  (SU),A      ;Sauvegarde du sujet
55      ; -----
56      ;Recherche de conditions
57 90C1 2A8190      LD  HL, (COND)
58 90C4 7E          LD  A, (HL)
59 90C5 328090      LD  (NCOND),A      ;Nombre de conditions
60      R2:      EQU  $      ;Boucle de recherche
61 90C8 3A7F90      LD  A, (SALLE)
62 90CB 23          INC  HL
63 90CC BE          CP   (HL)      ;Salle courante ?
64 90CD 2024        JR   NZ,R3      ;Non
65 90CF 23          INC  HL
66 90D0 3A8B90      LD  A, (VE)
67 90D3 BE          CP   (HL)      ;Verbe courant ?

```

```

68 90D4 201E          JR   NZ,R4           ;Non
69 90D6 23           INC   HL
70 90D7 3A8790        LD   A,(SU)
71 90DA BE           CP   (HL)           ;Sujet courant ?
72 90DB 2018          JR   NZ,R5           ;Non
73                   ;A ce pt, un contexte est trouve
74 90DD 23           INC   HL
75 90DE 7E           LD   A,(HL)
76 90DF 328990        LD   (CH),A         ;Changement de salle
77 90E2 23           INC   HL
78 90E3 7E           LD   A,(HL)
79 90E4 328A90        LD   (RE),A         ;Reponse
80 90E7 23           INC   HL
81 90E8 7E           LD   A,(HL)
82 90E9 328B90        LD   (PT),A         ;Points
83 90EC 23           INC   HL
84 90ED 7E           LD   A,(HL)
85 90EE 328C90        LD   (FI),A         ;Fin de partie
86 90F1 180F          JR   FIN            ;Fin du S/P
87 90F3 23           R3:  INC   HL
88 90F4 23           R4:  INC   HL
89 90F5 23           R5:  INC   HL
90 90F6 23           INC   HL
91 90F7 23           INC   HL
92 90F8 23           INC   HL
93 90F9 3A8090        LD   A,(NCOND)
94 90FC 3D           DEC   A
95 90FD 328090        LD   (NCOND),A     ;Nombre de conditions
96 9100 20C6          JR   NZ,R2         ;Boucle de recherche
97                   ;
98                   FIN:  EQU  $
99 9102 C9           RET
100                  END

```

Décomposition du programme

Initialisation	Lignes 33 à 39
Recherche de verbe	Lignes 41 à 47
Recherche de sujet	Lignes 48 à 54
Recherche de conditions	Lignes 57 à 99